In[29]:=
```
(*
This function runs the iteration for a given starting value.
       inputs:
         r = r value in the text,
    x0 = initialx value,
    n = number  of iterations,
    nreturn = number  of values to return, from  the end of the list generated
*)
run[r_, x0_, n_, nreturn_] := Module[{x}, x = Table[0, {n}];
    x[[1]] = x0;
    For[i = 1, i < n, i++, x[[i+1]] = r x[[i]] (1 - x[[i]])];
    Take[x, -nreturn]]
```

In[30]:=
```
(*
The range of r values is specified by r1,r2,dr.
*)
rplotProblem1 [r1_, r2_, dr_] := ListPlot[
    Transpose[
      Table[run[r, 0.01, 1000, 4], {r, r1, r2, dr}]
    ]
    , PlotMarkers → {Automatic , 8},
    DataRange → {r1, r2}, PlotRange → All, AxesLabel → {"r", "x"}]
```

In[31]:=
```
(* The experimental  dots *)
experimentalDots  = rplotProblem1  [0, 4, 0.1]
```

Out[31]=



In[32]:=
```
(* This is Equation (12.43)  *)
f[x_] := r x (1 - x)
```

In[33]:=
```
(* Find the fixed points *)
fixedPoints= Solve[f[x] == x, x]
```

Out[33]= $\left\{\{x \to 0\}, \left\{x \to \frac{-1+r}{r}\right\}\right\}$

In[34]:=
```
(*  Note we use '=' rather than ':='  *)
fixedPointA[r_] = x /. (fixedPoints[[1]])
```

Out[34]= 0

In[35]:=
```
fixedPointB[r_] = x /. (fixedPoints[[2]])
```

Out[35]= $\frac{-1+r}{r}$

In[36]:=
```
p = Factor[f[f[x]] - x]
```

Out[36]= $-x(1-r+r\,x)(1+r-r\,x-r^2\,x+r^2\,x^2)$

In[37]:=
```
gFixedPoints = Solve[p == 0, x] // Simplify
```

Out[37]= $\left\{\{x\to 0\},\ \left\{x\to \frac{-1+r}{r}\right\},\ \left\{x\to \frac{1+r-\sqrt{-3-2\,r+r^2}}{2\,r}\right\},\ \left\{x\to \frac{1+r+\sqrt{-3-2\,r+r^2}}{2\,r}\right\}\right\}$

In[38]:=
```
twoCyclePointC[r_] = x /. (gFixedPoints[[3]])
```

Out[38]= $\frac{1+r-\sqrt{-3-2\,r+r^2}}{2\,r}$

In[39]:=
```
twoCyclePointD[r_] = x /. (gFixedPoints[[4]])
```

Out[39]= $\frac{1+r+\sqrt{-3-2\,r+r^2}}{2\,r}$

In[40]:=
```
(r + 1) (r - 3) // Expand
```

Out[40]= $-3-2\,r+r^2$

In[41]:=
```
(* A quick look at the four functions, without the dashing  *)
Plot[{fixedPointA[r], fixedPointB[r], twoCyclePointC[r], twoCyclePointD[r]},
  {r, 0, 4}, PlotRange → {0, 1}, PlotStyle → Thickness[0.01], AxesLabel → {"r", "x"}]
```

Out[41]=



In[42]:=
```
ps1 = Thickness[0.005]; ps2 = {Thickness[0.005], Dashing[{0.005, 0.01}]};
```

In[43]:=
```
plotAstable = Plot[fixedPointA[r], {r, 0, 1}, PlotStyle → ps1];
```

```
In[44]:= plotAunstable = Plot[fixedPointA[r], {r, 1, 4}, PlotStyle → ps2];

In[45]:= plotBstable = Plot[fixedPointB[r], {r, 1, 3}, PlotStyle → ps1];

In[46]:= plotBunstable = Plot[fixedPointB[r], {r, 3, 4}, PlotStyle → ps2];

In[47]:= f'[x] // Factor (* bottom  of page 505 *)

Out[47]= -r (-1 + 2 x)

In[48]:= gprime  = f'[twoCyclePointC[r]] f'[twoCyclePointD[r]] // Simplify (* Eq. (12.57)  *)

Out[48]= 4 + 2 r - r^2

In[49]:= r1 = r /. First@Solve[{gprime  == 1, r > 0}, r]

Out[49]= 3

In[50]:= r2 = r /. First@Solve[{gprime  == -1, r > 0}, r]

Out[50]= 1 + Sqrt[6]

In[51]:= plotCstable = Plot[twoCyclePointC[r], {r, r1, r2}, PlotStyle → ps1];

In[52]:= plotDstable = Plot[twoCyclePointD[r], {r, r1, r2}, PlotStyle → ps1];

In[53]:= plotCunstable = Plot[twoCyclePointC[r], {r, r2, 4}, PlotStyle → ps2];

In[54]:= plotDunstable = Plot[twoCyclePointD[r], {r, r2, 4}, PlotStyle → ps2];

In[55]:= (* The theoretical curves *)
         theoreticalCurves = Show[plotAstable, plotAunstable,
             plotBstable, plotBunstable, plotCstable, plotCunstable,
             plotDstable, plotDunstable, PlotRange -> All, AxesLabel → {"r", "x"}]
```

Out[55]=



In[56]:=
```
Show[theoreticalCurves, experimentalDots ]
```

Out[56]=